```
$PROBLEM TWO COMPARTMENT ORAL                    ;DATE = 6/2/04 PROGRAMMER=XXXX
                                                 ;Units: Time=hr, Concentration=ug/ml,
                                                 ;Dose = 100mg or 250mg

$DATA ORAL2LAGEST_PAR.CSV IGNORE=C

$INPUT ID TIME CONC=DV AMT DOSE MDV

$SUBROUTINE ADVAN4 TRANS4

$PK
   CL  = THETA(1)*EXP(ETA(1))   ;CL/F
   V2  = THETA(2)*EXP(ETA(2))   ;V2/F
   Q   = THETA(3)*EXP(ETA(3))   ;Q/F
   V3  = THETA(4)*EXP(ETA(4))   ;V3/F
   KA  = THETA(5)*EXP(ETA(5))   ;Absorption rate constant
   ALAG1 = THETA(6)*EXP(ETA(6)) ;Lag time


   S2 = V2                      ;Scaling factor


$ERROR
   IPRED=F
   Y=F+F*ERR(1)+ERR(2)
                                ;Additive and proportional residual error model

$THETA   (0.01,1)         ;POPCL/F
$THETA   (1,10)           ;POPV2/F
$THETA   (0.01,1.5)       ;POPQ/F
$THETA   (0.11,5)         ;POPV3/F
$THETA   (0.1,3)          ;POPKA
$THETA   (0.1,1)          ;POPALAG1

$OMEGA   0.09            ;BSVCL/F
$OMEGA   0.09            ;BSVV2/F
$OMEGA   0.00,FIX        ;BSVQ/F
$OMEGA   0.09            ;BSVV3/F
$OMEGA   0.09            ;BSVKA
$OMEGA   0.09            ;BSVALAG1

$SIGMA  0.0,FIX        ;ERRCV
$SIGMA  1              ;ERRSD

$ESTIMATION METHOD=0 MAXEVAL=9999 PRINT=5 POSTHOC
$COVARIANCE MATRIX=S

$TABLE ID TIME DV IPRED DOSE
 NOPRINT ONEHEADER FILE=ORAL2LAGEST_PAR.FIT
```